
Journeymen’s Tools: Data Review Macro FreqAll: Using Proc SQL List
Processing with Dictionary.Columns to Eliminate Macro Do Loops

Ronald J. Fehd, Centers for Disease Control and Prevention, Atlanta, GA, USA

ABSTRACT

The SAS R© macro language is simple, yet powerful. List Processing
with Proc SQL is also simple, yet powerful. This paper provides a
data review macro FreqAll which illustrates using Proc SQL reading
Dictionary.Columns to replace macro %do loops.
Expected audience is advanced users, and macro programmers.

Keywords: data review, dynamic programming, list processing, do
loop, macro, SQL.

INTRODUCTION
Whenever I receive a data set, the first thing I want to do is examine
the data, looking first at the data structure, (Proc Contents) then at a
frequency listing of all the variables (Proc Freq). In data sets contain-
ing continuous variables the frequency listing gets long quickly. What
I really want is similar to Proc Univariate: a list of the 10 high and low
values. How can I make Proc Freq do that?
In this paper I will develop a macro with a macro do loop for each vari-
able using Dictionary.Columns and then replace that loop with Proc
SQL list processing.

The finished product will be a listing which contains:

1. data structure list similar to Proc Contents

2. abbreviated freqency showing the high and low values

PROC CONTENTS
There are several ways to access the data structure of a data set:

1. Proc Contents:
C-Contents.sas

1 Proc Contents data = SAShelp.Class;

2. Proc Datasets:

C-Datasets.sas
1 Proc Datasets library = SAShelp
2 details nolist
3 memtype = data;
4 contents
5 data = Class;
6 quit;

3. Proc Print:

C-Print.sas
1 PROC Print data = SAShelp.Vcolumn
2 (where = ( Libname eq ’SASHELP’
3 and MemName eq ’CLASS’
4 ) );

4. Proc SQL, describe:
C-SQL-describe.sas

1 PROC SQL; describe table SAShelp.Class
2 ; quit;

5. Proc SQL, select:

C-SQL-select.sas
1 PROC SQL; select Name, Type, Length, Label
2 from Dictionary.Columns
3 where Libname eq ’SASHELP’
4 and MemName eq ’CLASS’
5 ; quit;

I will be working with the SQL-select example, which will provide both
the data structure list and the loop of macro calls.

PROC FREQ
Proc Frequency provides a listing of all
values of a variable. For large data sets
with continuous variables, the listing gets
long quickly.

F-Freq.sas
1 PROC Freq data = SAShelp.Prdsal2;
2 tables _all_;

1800 lines =̃ 36 pages!

F-Freq.lst
1809 Month/Year
1810

1811 Cumulative Cumulative
1812 MONYR Frequency Percent Frequency Percent
1813 ----------------------------------------------------------
1814 JAN95 24 2.40 24 2.40
1815 FEB95 24 2.40 48 4.80

My goal is an output of few pages: one for data structure listing and
others for frequency listing.

PROC UNIVARIATE
Proc Univariate provides a listing of the
extreme values of a variable, but only nu-
merics.

F-Univariate.sas
1 PROC Univariate data = SAShelp.Class;
2 var _numeric_;

F-Univariate.lst snip 1
3 The UNIVARIATE Procedure
4 Variable: Age

. . .
F-Univariate.lst snip 2

52 Extreme Observations
53

54 ----Lowest---- ----Highest---
55

56 Value Obs Value Obs
57

58 11 18 15 8
59 11 11 15 14
60 12 16 15 17
61 12 13 15 19
62 12 10 16 15

The above examples illustrate my wish list: a list of variable attributes,
and a limited frequency listing, showing only the extreme values.
In the next section I will show a macro do loop, illustrate how to use it
for a procedure, and examine its programming problems.

MACRO DO LOOP
A do loop in a macro is similar to a
data step loop. On listing line 28, log
line 8, the index, I, is incremented from
the lower bound, 1, to the upper bound,
the macro variable Dim-Item. The macro
variable array, Item, contains 3 elements,
the sequentially numbered macro vari-
ables: Item1, Item2, and Item3. The
dimension of the macro array is Dim-
Item. This naming convention is neces-
sary in order for the loop to access each
element in the loop with the reference:
double ampersand, array-name, index —
&&Item&I.— shown in log line 9.

Macro-do-loop.log
21 1 %Macro Do_Loop;
22 2 %local Item1 Item2 Item3 Dim_Item I;
23 3 %Let Item1 = X1 ;
24 4 %Let Item2 = Y-2;
25 5 %Let Item3 = Z 3;
26 6 %Let Dim_Item = 3 ;
27 7
28 8 %Do I = 1 %to &Dim_Item.;
29 9 %Put Item&I<&&Item&I.>;
30 10 %end;
31 11 %mend;
32 12 %Do_Loop
33 Item1<X1>
34 Item2<Y-2>
35 Item3<Z 3>

The problems associated with using macro arrays are:

log line statement
ensure the scope of macro array variables 2 %local Item1 . . .
allocation of each element 3-5 %Let Item? = . . .
allocation of the dimension (upper bound) 6 %Let Dim Item = 3;

PROC SQL SELECT INTO
Fehd[Array-Contents] illustrates using a
Proc Contents output data set to create
a macro array. Fehd[Array-SQL] shows
how to use Proc SQL to do the same,
as shown here. Clay[Tight-Looping] pro-
vides comprehensive analysis of usage of
macro arrays. Whitlock[Macro-Design] re-
views macro design and list processing.

SQL-select-into.sas
1 PROC SQL noprint;
2 select Name
3 into :Name1 - :Name999
4 from Dictionary.Columns
5 where Libname eq ’SASHELP’
6 and MemName eq ’CLASS’
7 ;quit;
8 %Put _User_;

Note the system-generated automatic
macro variable SQLobs, listing line 35,
has the upper bound value, 5.

SQL-select-into.log
35 GLOBAL SQLOBS 5
36 GLOBAL SQLOOPS 38
37 GLOBAL NAME4 Height
38 GLOBAL NAME5 Weight
39 GLOBAL NAME2 Sex
40 GLOBAL NAME3 Age
41 GLOBAL NAME1 Name

MACRO FREQ-LOOP
In this section I will show a demonstration macro which makes a macro
array of the variables in a data set and then a macro do loop.
As noted above, I will use
Dictionary.Columns
to supply the listing of the data structure.

Describe-Dict-Columns.sas
1 PROC SQL; describe table Dictionary.Columns;
2 quit;

This table contains these columns:

Describe-Dict-Columns.log
28 libname char(8) label=’Library Name’,
29 memname char(32) label=’Member Name’,
30 memtype char(8) label=’Member Type’,
31 name char(32) label=’Column Name’,
32 type char(4) label=’Column Type’,
33 length num label=’Column Length’,
34 npos num label=’Column Position’,
35 varnum num label=’Column Number in Table’,
36 label char(256) label=’Column Label’,
37 format char(16) label=’Column Format’,
38 informat char(16) label=’Column Informat’,

Macro FreqAll has three parameters: li-
bref, data set name and how many ex-
treme values to show.

FreqAll-Loop.sas snip 1
4 %Macro FreqAll /* ---------------------- */
5 (In_Lib = SAShelp
6 ,In_Data = Class
7 ,ShowHiLo = 5

Note that two macro arrays are created —
Name and Type — in lines 29-30, and the
scope of their variables is declared in lines
25-26.

FreqAll-Loop.sas snip 2
25 %** Initialize macro array; %local SqlObs SqLoops;
26 %Do I = 1 %to 999; %local Var&I. Type&I.; %end;
27

28 PROC SQL noprint;
29 select Name, Type
30 into :Var1 - :Var999
31 ,:Type1 - :Type999
32 from Dictionary.Columns
33 where LibName eq "&In_Lib."
34 and MemName eq "&In_Data."
35 ;quit;

The loop begins on line 39 and ends on
line 56.

Note the four macro array element refer-
ences (&&Var&I.) in lines:

41
43
44 (&&Type&I.)
48

For this demonstration I make a report
with only the highest values. See the com-
plete high and low processing in macro
FreqOf below.

FreqAll-Loop.sas snip 3
39 %Do I = 1 %to &SQLobs.;%* ------------- *;
40 Proc Freq data = &In_Lib..&In_Data.;
41 tables &&Var&I.
42 / list missing noprint
43 out = Freq(rename = (&&Var&I. =
44 %If &&Type&I. eq char %then ValuC;
45 %Else ValuN;
46 ) );
47 DATA Freq;
48 retain Var "&&Var&I.";
49 if 0 then %** copy structure;
50 set FreqList;
51 set Freq;
52

53 PROC Append base = FreqList
54 data = Freq(obs = &ShowHiLo.)
55 force;
56 %end;%*do I ............................ *;
57

58 PROC Print data = FreqList;
59 by Var notsorted;
60 id Var;
61 run;%* ................................. *;%Mend;

FREQALL: SQL REPLACES MACRO DO LOOP
Where FreqAll was the name of the macro which contained the macro
array and do loop, here I name the program FreqAll and have placed
the statements inside the do loop into the macro subroutine FreqOf.

Note that the parameter names are
aligned with the variables from the Dic-
tionary.Columns data set: Name, Type,
Length, Format and Label.
If the user desires not the highest and low-
est values but the highest and lowest fre-
quencies, then I have provided a parame-
ter, Order = freq, which, will show the
mode: the values occuring most and least
often.
Note that the parameters InLib, InData,
and Nobs2View refer to global macro vari-
ables.

FreqAll.sas macro FreqOf pg. 1
43 %Macro Freqof
44 (Name = /* var */
45 ,Type = /* in (char,num) */
46 ,Length = /* integer */
47 ,Format = /* $char */
48 ,Label = /* $char40 */
49 /* for hi and low, use either: ****** */
50 ,Order = internal /* default: hi and low values */
51 /* for mode use: ** *** *** ****** **
52 ,Order = freq /* hi and low descending count */
53 ,InLib = &In_Lib. /* FreqAll scope: global */
54 ,InData = &In_Data. /* FreqAll scope: global */
55 ,Nobs2View = &HiLowToView /* FreqAll scope: global */
56 ,Testing = 0 /* show stuff? ,Testing=1 */
57 )/des = ’site: FreqOf: subroutine of prog FreqAll’
58 ;/*change notes

Compare to program Freq-All-Loop.

The Proc Freq is the same except for the
addition of the order = parameter. The
macro array references (&&Name&I.,
&&Type&I.) have been changed to pa-
rameter name (macro variable) refer-
ences: &Name, &Type.

Data Freq, attrib standardizes the
data set structure.

This section either appends a small listing

or divides the list into high and low sets
of values and adds a note indicating that
values were removed.

FreqAll.sas macro FreqOf pg. 2
65 %local Nobs; %Let Nobs = 0;%*initialize for symput;
66

67 PROC Freq data = &InLib..&InData.
68 order = &Order.;
69 tables &Name
70 / list missing noprint
71 out = Freq(rename = (&Name. =
72 %If &Type = char %then ValuC;
73 %else ValuN; ));
74

75 DATA Freq; %* name32 type4 $200 label40;
76 attrib Attributes length = $%eval(32 +1 +4 +1 +4 +1 + 40)
77 ValuC length = $20
78 ValuN length = 8
79 Count length = 4 format = comma.
80 Percent length = 8 format = 6.2
81 Level length = 4;
82

83 retain Attributes "&Name. &Type.:&Length. &Label"
84 ValuC "." ValuN Level .;
85

86 do until(EndoFile);
87 set Freq end = EndoFile
88 nobs = Nobs;
89 Level ++1; %*increment retained counter;
90 output; end; %*do until EndoFile;
91 call symput( ’Nobs’, compress(put(Nobs,32.)));
92 stop; run;
93 %If &Testing. %then %Put Note:&SysMacroName.: nobs<&Nobs.>;
94

95 %If &Nobs. le %eval(2 * &Nobs2View. + 2) %then %do;
96 PROC Append base = ListFreq
97 data = Freq; %end;
98

99 %Else %do;%* --------------------------- *;
100 DATA Snipped;
101 set Freq(obs = 1);ValuC = ’<snipped>’;ValuN = .;
102 Count = .; Percent = .;
103 Level = .; output; stop;
104

105 PROC Append base = ListFreq %*high values;
106 data = Freq(obs = &Nobs2View.);
107

108 PROC Append base = ListFreq
109 data = Snipped;
110

111 PROC Append base = ListFreq %*low values;
112 data = Freq(firstobs = %eval( &Nobs.
113 - &Nobs2View. +1));
114 %end;%*Else do ......................... *;
115 run;%* ................................. *; %Mend Freqof;

In order for the global macro variables
In Lib and In Data to be used in
the select ... from ... where
phrase, lines 128-129 and 133-134, each
must be in all caps.

Proc SQL creates three objects:

1. line 124: table ListAttributes con-
taining the variable attributes; this is
the first page of the summary report.

2. line 137: macro variable List con-
taining calls of macro FreqOf for
each variable; these are executed
on line 145.
To view the FreqOf statements, dis-
able line 121:
%*Let SQLprint = noprint;
This select statement, lines 131-
138, replaces the macro array and
do loop in the FreqAll-Loop pro-
gram.

3. line 139: macro variable NobsData:
the number of observations of the in-
put data set; this is used in the title2
statement, lines 148-149.

The report is printed in two parts: at-
tributes, and frequencies.

Housecleaning: delete the program’s
global macro variables.

FreqAll.sas List Processing
117 %Let In_Lib = %upcase(&In_Lib.);
118 %Let In_Data = %upcase(&In_Data.);
119

120 %Let SQLprint = print;%*testing: view FreqOf statements;
121 %Let SQLprint = noprint;
122

123 PROC SQL &SQLprint.;
124 create table ListAttributes as
125 select Varnum, Name, Type, Length
126 , Format, Informat, Label, Npos
127 from Dictionary.Columns
128 where Libname eq "&In_Lib."
129 and MemName eq "&In_Data."
130 and MemType eq ’DATA’
131 ; select ’%FreqOf(name = ’ !! trim(Name)
132 !! ’,type = ’ !! trim(Type)
133 !! ’,length = ’
134 !! compress(put(Length,32.))
135 !! ’,label = ’ !! trim(Label)
136 !! ’)’
137 into :List separated by ’ ’
138 from ListAttributes
139 ; select Nobs into :NobsData
140 from Dictionary.Tables
141 where Libname eq "&In_Lib."
142 and MemName eq "&In_Data."
143 and MemType eq ’DATA’
144 ; quit;
145 &List. ; %*execute macro calls;
146

147 %Let NobsData = &NobsData;%* remove leading blanks;
148 Title2
149 "&SysProcessName.: &In_Lib..&In_Data. nobs:&NobsData.";
150

151 PROC Print data = ListAttributes noobs;
152 Title3 "variable attributes";
153

154 PROC Print data = ListFreq;
155 var Valu: Count Percent Level;
156 by Attributes notsorted;
157 id Attributes ;
158 Title3 "list of variable frequencies";
159 run;
160 %symdel In_Data In_Lib HiLowToView List NobsData SQLprint;

Changing line 121 to:
%*Let SQLprint = noprint;
produces this output, which shows the
statements in the macro variable List.
Note: spaces have been added to align
columns and improve readability.

FreqAll-List-of-FreqOf.txt
1 Program FreqAll: SASHELP.CLASS nobs:19
2 %FreqOf(name = Name , type = char, length = 8, label = )
3 %FreqOf(name = Sex , type = char, length = 1, label = )
4 %FreqOf(name = Age , type = num , length = 8, label = )
5 %FreqOf(name = Height, type = num , length = 8, label = )
6 %FreqOf(name = Weight, type = num , length = 8, label = )

This is the FreqAll report for
SAShelp.PrdSal2; compare
to program F-Freq.sas.
The first page of the FreqAll
report contains Proc Con-
tents information.

FreqAll.txt PrdSal2 pg. 1
1 SUG-2006: data review utility program FreqAll
2 Program FreqAll: SASHELP.PRDSAL2 nobs:23040
3 variable attributes
4

5 varnum name type length format informat label npos
6

7 1 COUNTRY char 10 $CHAR10. Country 48
8 2 STATE char 22 $CHAR22. State/Province 58
9 3 COUNTY char 20 $CHAR20. County 80

10 4 ACTUAL num 8 DOLLAR12.2 Actual Sales 0
11 5 PREDICT num 8 DOLLAR12.2 Predicted Sales 8
12 6 PRODTYPE char 10 $CHAR10. Product Type 100
13 7 PRODUCT char 10 $CHAR10. Product 110
14 8 YEAR num 8 4. Year 16
15 9 QUARTER num 8 8. Quarter 24
16 10 MONTH num 8 MONNAME3. Month 32
17 11 MONYR num 8 MONYY. MONYY. Month/Year 40

The second page contains the abbrevi-
ated frequencies of each variable.

Note: the listing (.lst) was saved to text
(.txt) and the CR/LF removed for print-
ing here; the listing is truncated to save
space.
The complete listing from program FreqAll
is 95 lines; three pages, instead of 20 from

Proc Contents: 2 pages
Proc Freq: 18 pages

FreqAll.txt PrdSal2 pg. 2

21 list of variable frequencies
22
23 Attributes ValuC ValuN Count Percent Level
24
25 COUNTRY char:10 Country U.S.A. . 1,000 100.00 1
26
27 STATE char:22 State/Province California . 1,000 100.00 1
28
29 COUNTY char:20 County . 1,000 100.00 1
30
31 ACTUAL num:8 Actual Sales . 8.16 1 0.10 1
32 . 10.20 1 0.10 2
33 . 13.60 1 0.10 3
34 . 17.00 1 0.10 4
35 . 27.20 1 0.10 5
36 <snipped> . . . .
37 . 3410.20 1 0.10 851
38 . 3413.60 1 0.10 852
39 . 3415.30 1 0.10 853
40 . 3440.80 1 0.10 854
41 . 3515.60 1 0.10 855
42
43 PREDICT num:8 Predicted Sales . 4.08 2 0.20 1
44 . 5.10 2 0.20 2
45 . 9.52 1 0.10 3
46 . 11.90 1 0.10 4
47 . 16.32 3 0.30 5
48 <snipped> . . . .
49 . 3593.80 1 0.10 844
50 . 3604.00 2 0.20 845
51 . 3605.70 1 0.10 846
52 . 3609.10 1 0.10 847
53 . 3634.60 1 0.10 848

Fehd in [Prog-Doc] dis-
cusses necessary items in
a program header.

FreqAll.sas documentation
1 /* Name: FreqAll.sas
2

3 Requirements : description : Proc Freq of all vars in data set
4 purpose : provide shorter listing than tables _all_
5

6 ------------------------------------------------------------------------
7 Contexts : program group: data review
8 program type: routine
9 SAS type: program with parameters

10 uses routines: in-program macro %FreqOf
11

12 ------------------------------------------------------------------------
13 Specifications: input : libref
14 data
15 max number of high and low obs to view
16 process: SQL writes macro calls
17 Proc Freq of each var
18 save to data
19 append to report
20 subset if Nobs greater than Max-N-to-view
21 output : print report
22

23 ------------------------------------------------------------------------
24 Information : author: Ronald J. Fehd
25

26 ------------------------------------------------------------------------
27 Usage Examples:
28 %Let In_Lib = Library;
29 %Let In_Data = MyData ;
30 %Let HiLowToView = 5 ;
31

32 %Inc ’FreqAll.sas’;

To receive the latest edition of this pro-
gram header send an e-mail to the au-
thor with the subject:
request FreqAll

SUGGESTED READINGS
Wobus and Gober[Data-Analysis] show data review with procs Summary and Univariate. Abolafia[Data-Check]
provides a macro to reproduce Proc DataChk.

CONCLUSION
FreqAll The data review utility program FreqAll provides a shorter data set

summary with more information.
Proc SQL List processing (select . . . into :List) can eliminate the use of macro

arrays and do loops. This yields clearer code.

